

雙田 悠平 2019/03/21

プログラム (円柱周りの流れ)

```
int n1 = 30;
int n2 = 60;
real nu = 1.0/1000.0;
real dt = 0.05;
real alpha = 1.0/dt;
int timestepmax = 400;
real eps = 0;
string dirname, fname;

border ba(t=0, 1.0) {x=t*10.0-1.0;y=-1.0;label=1;};
border bb(t=0, 1.0) {x=9.0;y=2.0*t-1.0;label=2;};
border bc(t=0, 1.0) {x=9.0-10.0*t;y=1.0;label=3;};
border bd(t=0, 1.0) {x=-1.0;y=1.0-2.0*t;label=4;};
border cc(t=0, 2*pi) {x=cos(t)*0.25+0.75;y=sin(t)*0.25;label=5;};

mesh Th=buildmesh(ba(n2)+bb(n1)+bc(n2)+bd(n1)+cc(-n1));

plot(Th);

fespace Xh(Th, [P2, P2, P1]);
fespace Vh(Th, P2);
fespace Qh(Th, P1);

Xh [u1, u2, p], [v1, v2, q];
Vh up1, up2, w, ww;
Qh pp;

macro d11(u1) dx(u1)//
macro d22(u2) dy(u2)//
```

```
macro d12(u1, u2) (dy(u1) + dx(u2))/2.0//
```

```
macro div(u1, u2) (dx(u1) + dy(u2))//
```

```
Qh psi, phi;
```

```
func stinflow=y-y*y*y/3.0;
```

```
problem streamlines(psi, phi, solver=UMFPACK) =  
    int2d(Th) ( dx(psi)*dx(phi) + dy(psi)*dy(phi))  
+ int2d(Th) ( phi*(dy(u1)-dx(u2)))  
+ on(1, psi=(-2.0/3.0))  
+ on(4, psi=stinflow)  
+ on(3, psi=(2.0/3.0))  
+ on(5, psi=0.0);
```

```
streamlines;
```

```
plot(psi, wait=1);
```

```
problem Stokes([u1, u2, p], [v1, v2, q], solver=UMFPACK) =  
    int2d(Th) (+ 2.0*nu * (d11(u1)*d11(v1)+2.0*d12(u1, u2)*d12(v1, v2)+d22(u2)*d22(v2))  
    - p * div(v1, v2) - q * div(u1, u2))  
+ on(1, 3, u2=0)  
+ on(4, u1=1.0, u2=0)  
+ on(5, u1=0, u2=0);
```

```
int i;
```

```
real D, L;
```

```
problem NS([u1, u2, p], [v1, v2, q], solver=UMFPACK, init=i) =  
    int2d(Th) ( alpha * (u1*v1 + u2*v2)  
+ 2.0*nu * (d11(u1)*d11(v1)+2.0*d12(u1, u2)*d12(v1, v2)+d22(u2)*d22(v2))  
- p * div(v1, v2) - q * div(u1, u2))
```

```

- int2d(Th) ( alpha * (convect ([up1, up2], -dt, up1)*v1
+convect ([up1, up2], -dt, up2)*v2) )
+ on(1, 3, u2=0)
+ on(4, u1=1.0, u2=0)
+ on(5, u1=0, u2=0);

```

```

solve Laplace(w, ww)
= int2d(Th) (dx(w)*dx(ww)+dy(w)*dy(ww))
+on(1, 2, 3, 4, w=0)
+on(5, w=1);

```

```

int count=0; {
    ofstream outputfile("tdl400_"+eps+".txt");
    dirname = "graph_" + eps;
    system("mkdir " + dirname);

```

```

plot(w, wait=1);

```

```

real area = int2d(Th) (1.0);
real meanp;

```

```

plot(Th, wait=1);

```

```

for ( i = 0; i < timestepmax; i++) {
    up1=u1;
    up2=u2;
    NS;
    D=int2d(Th) (//抗力
w*(u1-up1)/dt//(au/at, vD)
+(u1*d11(u1)+u2*d22(u1))*w//a1(u, u, vD)
+2.0*nu * (d11(u1)*d11(w)+2.0*d12(u1, u2)*dy(w)/2.0)//a(u, vD)
+dx(w)*p//b(vD, p)
);

```

```

L=int2d(Th) (//揚力
    w*(u2-up2)/dt/(au/at, vL)
    +(u1*d11(u2)+u2*d22(u2))*w//a1(u, u, vL)
    +2.0*nu * (2.0*d12(u1, u2)*dx(w)/2.0+d22(u2)*d22(w))//a(u, vL)
    +dy(w)*p//b(VL, p)
);
meanp = int2d(Th) (p) / area;
outputfile<<dt*(i+1)<<" "<<L<<" "<<D<<endl;

    if ((i+1)%2 == 0) { //10 ステップごとに表示
        // ファイル名を決める。
        if (count < 10)
            fname=dirname+"/zu00"+count+".ps";
        else if (count < 100)
            fname=dirname+"/zu0"+count+".ps";
        else
            fname=dirname+"/zu"+count+".ps";
        count++;
    }

plot([u1, u2], p, ps=fname, wait=0, value=true, coef=0.1*10.0, cmm="t="+dt*(i+1));

        cout<<"L="<<L<<" , D="<<D<<endl;

    }
}
}

```

数値実験結果（上記プログラム実行結果）

```

-- mesh: Nb of Triangles = 2730, Nb of Vertices 1470
-- Solve :
    min -0.666667 max 0.666667
-- Solve :


```

```
min -1.18517e-37 max 1
exec: mkdir graph_0
mkdir: graph_0: File exists
-- Solve :
min -0.992452 max 204.474
-- Solve :
min -1.5811 max 1.8938
L=-0.00025072,D=0.216824
-- Solve :
min -1.53705 max 1.76828
-- Solve :
min -1.48406 max 1.68932
L=-0.000641968,D=0.181965
-- Solve :
min -1.4492 max 1.65492
-- Solve :
min -1.41463 max 1.63089
. . .
L=0.105407,D=0.770457
-- Solve :
min -1.16322 max 1.64296
-- Solve :
min -1.21008 max 1.59204
L=0.157446,D=0.773372
-- Solve :
min -1.25264 max 1.6165
-- Solve :
min -1.28751 max 1.61303
L=0.19537,D=0.776148
times: compile 0.009568s, execution 113.545s, mpirank:0
##### We forget of deleting 0 Nb pointer, 0Bytes , mpirank 0, memory leak
=4905664
CodeAlloc : nb ptr 3461, size :381832 mpirank: 0
```

Ok: Normal End

可視化は次のようになった。


初期状態の流れ関数 φ の等高線（流線）の表示



初期状態のラプラス方程式 w の等高線の表示




領域 Ω の三角形による分割の表示




t = 5における流線と流速ベクトル場の矢印（有向線分）表示と圧力の等高線表示

t=5




t = 8における流線と流速ベクトル場の矢印（有向線分）表示と圧力の等高線表示


t=8



t = 11.9における流線と流速ベクトル場の矢印（有向線分）表示と圧力の等高線表示



t = 20における流線と流速ベクトル場の矢印（有向線分）表示と圧力の等高線表示



参考文献

- ・野地 誠、流体中に置かれた回転円柱の運動、2008年度 龍谷大学理工学研究科修士論文 (2009)
- ・田所 大樹、流体中に置かれた回転円柱の運動、2017年度 明治大学総合数理学部卒業レポート (2018)
- ・田端 正久、流体問題の有限要素解析、数理科学 No.417、P.13-19、(1998 MARCH)
- ・大塚厚二、高石武史、有限要素法で学ぶ現象と数理、共立出版、P.144-145、(2014)
- ・竹内 淳、高校数学でわかる流体力学 ベルヌーイの定理から翼に働く揚力まで、ブルーバックス、講談社 (2014)

まとめ

卒業レポート作成を行い、流体中に置かれた円柱の運動を **FreeFem++**を用いての可視化とカルマン渦の存在を確認し、その際に使用したナビエ・ストークス方程式について、無次元化とレイノルズ数は理解することが出来ました。また、一般的な抗力と揚力の導き方の理解も出来ました。また領域 Ω の流入部の関数を変化させると、カルマン渦と円柱との距離が変わることが分かりました。

今後の課題としては、弱形式と精度よく抗力と揚力を求めることについての十分な理解、非定常の場合のレイノルズ数と抗力・揚力係数の関係を調べる際は周期的になった際の平均値にすることが挙げられます。